
Abstract

1 Introduction

A standard domain description language, PDDL (Ghallab et al. 1998; McDermott 2000; Fox and Long
2003), for deterministic planning domains has simplified sharing of domain models and problems in the
planning community, and has enabled direct comparisons of different planning systems. As a result, there
has been considerable progress in planning research with deterministic domain models since the first Inter-
national Planning Competition in 1998.

The 4th International Planning Competition included a probabilistic track for the first time in an attempt
to create a common platform for the evaluation of probabilistic and decision theoretic planning systems.
This document briefly describes the input language, PPDDL1.0, that was used for the probabilistic track.
PPDDL1.0 is essentially a syntactic extension of levels 1 and 2 of PDDL2.1 (Fox and Long 2003). Section 2
gives a comprehensive introduction to PDDL so that prior knowledge of PDDL2.1, or any of its predeces-
sors, is not required in order to understand this document. Readers already familiar with PDDL can jump
immediately to Section 3, which focuses on new language features specific to PPDDL, including proba-
bilistic effects and rewards. Section 4 provides a semantics for PPDDL1.0 planning problems in terms of
Markov decision processes(MDPs; Howard 1960; Puterman 1994). The complete syntax for PPDDL1.0 is
given in Appendix A.

Note that, unlike PDDL2.1, PPDDL1.0 does not impose a specific structure on plans, except that only a
single action can be executed at any point in time. The problem of plan representation has been left entirely
to the planning systems, and planning systems may even choose to have no plan representation at all. At the
planning competition, planning systems were evaluated using a client-server model where the server sent
states to a client (planning system) and the client was expected to respond with an action to execute for each
state it received.

PPDDL1.0 is a first step towards a general language for describing probabilistic and decision theoretic
planning problems. Directions for future extensions include, but are not limited to, concurrency, observabil-
ity, and time. Of these, concurrency is the most straightforward to add, requiring only a slightly modified
semantics in line with that proposed by Mausam and Weld (2004). Adding a temporal dimension to ac-
tions would permit the modeling of semi-Markov decision processes (Howard 1971), or in the presence
of concurrency, generalized semi-Markov decision processes (Younes and Simmons 2004). Restrictions in
the observability of state features would be necessary to model contingent probabilistic planning problems
(Draper et al. 1994), as well as partially observable MDPs (Åström 1965).

2 PDDL Basics

PDDL is a language for specifying deterministic planning domains and problems. We describe the basic
building blocks of PDDL in this section. In the next section, we introduce extensions necessary to express
probabilistic and decision theoretic planning domains and problems.

2.1 Planning Domains

A PDDL planning domain consists of a setT of types, a subtyping relationST ⊂ T × T , a setC of global
objects (domainconstants), a setP of predicates, a setF of functions, and a setAS of action schemata.
Predicates and functions are used to encodestate variables.

When defining a domain in PDDL, it is given a unique name that is used when referring to the domain
in problem definitions (see Section 2.2). Figure 1 shows the definition of a domain named “test-domain” in

(define (domain test-domain)
(:requirements :typing :equality :conditional-effects

:fluents)
(:types car box)
(:constants goldie - car)
(:predicates (parked ?x - car) (holding ?x - box)

(in ?x - box ?y - car))
(:functions (fuel-level ?x - car))
(:action load

:parameters (?x - box ?y - car)
:precondition (and (holding ?x) (parked ?y))
:effect (and (in ?x ?y)

(forall (?z - car)
(when (not (= ?z ?y))

(not (in ?x ?z))))))
(:action refuel

:parameters (?x - car)
:precondition (< (fuel-level ?x) 10)
:effect (increase (fuel-level ?x) 1)))

Figure 1: Definition of a simple planning domain in PDDL.

PDDL. The statement

(:requirements :typing :equality :conditional-effects :fluents)

signals to a planner reading the domain definition that support for typing, equality, conditional effects, and
fluents are required in order to correctly handle the domain being defined. Requirements are explained in
more detail in Appendix A, where a full grammar for PPDDL is provided. Tokens starting with a question
mark, for example?x , arevariables, which are not to be confused withstatevariables.

2.1.1 Types

Objects and variables areterms, and in PDDL all terms have some typeτ ∈ T . The domain definition in
Figure 1 declares two types:car andbox . The constantgoldie is declared to be of typecar , while the
first parameter of the action schemaload is declared to be of typebox .

All declared types are by default subtypes of the built-in PDDL typeobject . A type is also a subtype
of itself (the subtyping relationST is reflexive), and ifτ1 is a subtype ofτ2 and τ2 is a subtype ofτ3,
thenτ1 is a subtype ofτ3 (ST is transitive). For the example domain we haveT = {object , car , box }
with ST containing the elements〈object , object 〉, 〈car , object 〉, 〈car , car 〉, 〈box , object 〉,
and〈box , box 〉.

It is possible to specify the supertype of a type in the type declaration of a domain definition. For
example, the type declaration

(:types car - object saab volvo - car)

declares the typecar as a subtype ofobject , and the typessaab andvolvo as subtypes ofcar .

The typescar andobject are examples ofsimpletypes. PDDL also includes support forunion types
τ1 ∪ · · · ∪ τn, with the restriction that eachτi is a simple type.1 As an example of the use of union types,
consider the declaration

(:constants herbie - (either saab volvo))

of a domain constantherbie of typesaab ∪ volvo . A union typeτ1 ∪ · · · ∪ τn is a subtype ofτ if τi is
a subtype ofτ for all i ∈ {1, . . . , n}. On the other hand,τ is a subtype of a union typeτ1 ∪ · · · ∪ τn if τ is
a subtype ofτi for somei ∈ {1, . . . , n}.

2.1.2 Predicates and Functions

In PDDL, predicates are used to encode Boolean state variables, while functions are used to encode numeric
state variables. A function declaration in PDDL, such as

(:functions (fuel-level ?x - car))

in the example domain (Figure 1), is in reality a declaration of a function from PDDL objects to numeric
state variables. A predicate is also a function with PDDL objects as domain, but with boolean state variables
as range. The type of a function (predicate) parameter restricts the domain of the function (predicate). The
function fuel-level , for example, only applies to objects of typecar or a subtype ofcar .

The value of the application(parked goldie) is a boolean state variableparked goldie. Say that in
addition to the domain constantgoldie , there is also an objectups-box of typebox . Given this set of
objects, the predicates and functions of “test-domain” give rise to a state space made up of the following
state variables:

Name Type
parkedgoldie boolean
holdingups-box boolean
inups-box,goldie boolean
fuel-levelgoldie numeric

A function that does not take any arguments represents a single state variable with the same name as the
function. It is then possible to refer directly to the state variable in PDDL domain and problem definitions
without having to use function application. For example, given a0-ary functionscore , the formulas(=
(score) 17) and(= score 17) are equivalent.

It is worth noting that a domain definition alone does not, in general, determine the extent of the state
space for planning problems linked to the domain, unless all functions and predicates take no arguments. In
addition to objects declared as domain constants, objects can also be declared in problem definitions. Only
when the complete set of objects for a planning problem is known can the state space be determined.

2.1.3 Actions

Actions in PDDL can be thought of as representing sets of state transitions, with a state being a particular
assignment to the set of state variables of a planning problem. An action consists of a precondition, charac-
terizing the set of states that the action is applicable in, and an effect. The effect specifies updates to state
variables that occur at the execution of the action.

1The original version of PDDL (Ghallab et al. 1998) allowed unions of union types. Later versions, including PDDL2.1 (Fox
and Long 2003) only allow unions of simple types. The restriction is not significant as for any type that is a union of union types,
we can find an equivalent type that is a union of simple types by “flattening” the union.

Basic effects specify updates to individual state variables. For a boolean state variableb, the effectb (or
an application yielding the state variableb) simply means thatb should be set to true in the next state. To
setb to false, the notation(not b) is used. For a numeric state variablex, the general form for updates is
( 〈assign-op〉 x 〈f-exp〉) , where〈assign-op〉 is one ofassign , scale-up , scale-down , increase ,
or decrease , and〈f-exp〉 is a numeric expression.

Effects can be combined by using conjunction. PDDL also includes support for conditional effects of the
form (when c e) , meaning that the effecte only occurs in states satisfying the conditionc, and universally
quantified effects. Examples of all these kinds of effects are given in Figure 1.

An action schema, in the definition of a domain, declares a function from PDDL objects to actions, much
in the same way as functions and predicates are functions from PDDL objects to state variables. The action
schema

(:action refuel
:parameters (?x - car)
:precondition (< (fuel-level ?x) 10)
:effect (increase (fuel-level ?x) 1))

in “test-domain”, when applied to the objectgoldie , returns an actionrefuelgoldie. This action is applicable
in states satisfying the conditionfuel-levelgoldie < 10 and has the effect thatfuel-levelgoldie is increased
by one. It is considered an error to apply an action in a state where the precondition does not hold (cf. Fox
and Long 2003). If we want therefuelgoldie actions to be universally applicable, but with no state change
occurring iffuel-levelgoldie < 10 does not hold, then we should use conditional effects:

(:action refuel
:parameters (?x - car)
:effect (when (< (fuel-level ?x) 10)

(increase (fuel-level ?x) 1)))

2.2 Planning Problems

A planning problem consists of a set of state variablesV , a set of actionsA, an initial states0, a goal
condition φ identifying a set of goal states, and an optimization metricf that is typically a function of
numeric state variables evaluated in a goal state. A state is simply an assignment of values to the set of state
variables.

In PDDL, a planning problem is always associated with a domain definition, and the definition of a
planning problem includes a declaration of a set of problem-specific objectsO. The state variablesV for
the planning problem are obtained fromO, C, P , andF as all type-consistent applications of predicates
or functions to objects (including domain constants). The setA of actions is obtained similarly as all type-
consistent applications of action schemata inAS to objects inO ∪ C. The process of obtaining all state
variables and actions for a planning problem through the exhaustive application of predicates, functions,
and action schemata to objects is referred to asgrounding.

Figure 2 shows the definition of a simple planning problem associated with a domain named “test-
domain” (defined in Figure 1). The problem definition declares two problem-specific objects,ups-box
andcereal-box , both of typebox . The setV of state variables for this planning problem is listed in
Table 1. The table also shows the value of each state variable in the initial states0 of the problem, as
specified by(:init . . . ) in the problem definition. Note that boolean state variables not mentioned in
the init specification, for exampleholding cereal-box, are assumed to be false in the initial state (closed world
assumption). The actions for “test-problem” areloadups-box,goldie, loadcereal-box,goldie, andrefuelgoldie.

(define (problem test-problem)
(:domain test-domain)
(:objects ups-box cereal-box - box)
(:init (parked goldie)

(holding ups-box)
(in cereal-box goldie)
(= (fuel-level goldie) 7))

(:goal (and (in ups-box goldie) (>= (fuel-level goldie) 9))))

Figure 2: Definition of a simple planning problem in PDDL associated with the domain named “test-domain” defined
in Figure 1. Note that a hyphen (“- ”) can be part of a name for objects, types, etc., but that it is also used in the
assignment of types to objects and variables. A hyphen is assumed to be part of a name token, unless it is preceded
by white space. For example,cereal-box is a single name token, whilecereal-box - box specifies that
cereal-box has typebox .

Name Type Init
parkedgoldie boolean true
holdingups-box boolean true
holdingcereal-box boolean false
inups-box,goldie boolean false
incereal-box,goldie boolean true
fuel-levelgoldie numeric 7

Table 1: State variables and their initial values for “test-problem” defined in Figure 2.

3 Probabilistic and Decision Theoretic Extensions

We now describe syntactic extensions to PDDL that allows us to specify Markov decision processes (MDPs).
The key extension is support for probabilistic effects. Rewards, an essential part of MDPs, are modeled using
an existing language feature: fluents. However, we restrict the use of rewards so that full support for fluents
does not become a prerequisite for MDP planning.

3.1 Probabilistic Effects

In order to define probabilistic and decision theoretic planning problems, we need to add support for proba-
bilistic effects. The syntax for probabilistic effects is

(probabilistic p1 e1 . . . pk ek)

meaning that effectei occurs with probabilitypi. We require that the constraintspi ≥ 0 and
∑k

i=1 pi = 1
are fulfilled: a probabilistic effect declares an exhaustive set of probability-weighted outcomes. We do,
however, allow a probability-effect pair to be left out if the effect is empty. In other words, the effect
(probabilistic p1 e1 . . . pl el) with

∑l
i=1 pi < 1 is syntactic sugar for(probabilistic p1

e1 . . . pl el q (and)) with q = 1 − ∑l
i=1 pi. For example, the effect(probabilistic 0.9

(clogged)) means that with probability0.9 the state variableclogged becomes true in the next state,
while with probability0.1 the state remains unchanged. Outcomes are not required to be mutually exclusive.

Figure 3 shows an encoding in PPDDL of the “Bomb and Toilet” example described by Kushmerick
et al. (1995). The requirements flag:probabilistic-effects signals that probabilistic effects are

(define (domain bomb-and-toilet)
(:requirements :conditional-effects :probabilistic-effects)
(:predicates (bomb-in-package ?pkg) (toilet-clogged)

(bomb-defused))
(:action dunk-package

:parameters (?pkg)
:effect (and (when (bomb-in-package ?pkg)

(bomb-defused))
(probabilistic 0.05 (toilet-clogged)))))

(define (problem bomb-and-toilet)
(:domain bomb-and-toilet)
(:requirements :negative-preconditions)
(:objects package1 package2)
(:init (probabilistic 0.5 (bomb-in-package package1)

0.5 (bomb-in-package package2)))
(:goal (and (bomb-defused) (not (toilet-clogged)))))

Figure 3: PPDDL encoding of “Bomb and Toilet” example.

Name Type Init 1 Init 2
bomb-in-packagepackage1 boolean true false
bomb-in-packagepackage2 boolean false true
toilet-clogged boolean false false
bomb-defused boolean false false

Table 2: State variables and their initial values for the “Bomb and Toilet” problem.

used in the domain definition. In this problem, there are two packages, one of which contains a bomb. The
bomb can be defused by dunking the package containing the bomb in the toilet. There is a0.05 probability
of the toilet becoming clogged when a package is placed in it.

The problem definition in Figure 3 also shows that initial conditions in PPDDL can be probabilistic. In
this particular example, we define two possible initial states with equal probability (0.5) of being the true
initial state for any given execution. Table 2 lists the state variables for the “Bomb and Toilet” problem
and their values in the two possible initial states. Intuitively, we can think of the initial conditions of a
PPDDL planning problem as being the effects of an action forced to be scheduled right before time0.
Also, note that the goal of the problem involves negation, which is why the problem definition declares the
:negative-preconditions requirements flag.

PPDDL allows arbitrary nesting of conditional and probabilistic effects. This is in contrast to popular
propositional encodings, such as probabilistic STRIPS operators (PSOs; Kushmerick et al. 1995) and fac-
tored PSOs (Dearden and Boutilier 1997), which do not allow conditional effects nested inside probabilistic
effects. While arbitrary nesting does not add to the expressiveness of the language, it can allow for expo-
nentially more compact representations of certain effects given the same set of state variables and actions
(Rintanen 2003). Any PPDDL action can, however, be translated into asetof PSOs with at most a polyno-
mial increase in the size of the representation. Consequently, it follows from the results of Littman (1997)
that PPDDL,after grounding(i.e. full instantiation of action schemata), is representationally equivalent to

dynamic Bayesian networks (Dean and Kanazawa 1989), which is another popular representation for MDP
planning problems.

Still, it is worth noting that a single PPDDL actionschemacan represent a large number of actions and
a single predicate can represent a large number of state variables, meaning that PPDDL often can represent
planning problems more succinctly than other representations. For example, the number of actions that can
be represented usingm objects andn action schemata with arityc is m · nc, which is not bounded by any
polynomial in the size of the original representation(m + n). Grounding is by no means a prerequisite
for PPDDL planning, so planners could conceivably take advantage of the more compact representation by
working directly with action schemata.

3.2 Rewards

Markovian rewards, associated with state transitions, can be encoded using fluents. PPDDL reserves the
fluent reward , accessed as(reward) or reward , to represent the total accumulated reward since the
start of execution. Rewards are associated with state transitions through update rules in action effects. The
use of thereward fluent is restricted to action effects of the form( 〈additive-op〉 〈reward fluent〉 〈f-exp〉) ,
where〈additive-op〉 is eitherincrease or decrease , and〈f-exp〉 is a numeric expression not involving
reward . Action preconditions and effect conditions are not allowed to refer to thereward fluent, which
means that the accumulated reward does not have to be considered part of the state space. The initial value
of reward is zero. These restrictions on the use of thereward fluent allow a planner to handle domains
with rewards without having to implement full support for fluents.

A new requirements flag,:rewards , is introduced to signal that support for rewards is required. Do-
mains that require both probabilistic effects and rewards can declare the:mdp requirements flag, which
implies :probabilistic-effects and:rewards .

Figure 4 shows part of the PPDDL encoding of a coffee delivery domain described by Dearden and
Boutilier (1997). A reward of0.8 is awarded if the user has coffee after the “buy-coffee” action has been
executed, and a reward of0.2 is awarded ifis-wet is false after execution of “buy-coffee”. Note that a total
reward of1.0 can be awarded as a result of executing the “buy-coffee” action if execution of the action leads
to a state where bothuser -has-coffee and¬is-wet hold.

3.3 Plan Objectives

Regular PDDL goals are used to express goal-type performance objectives. A goal statement(:goal
φ) for a probabilistic planning problem encodes the objective that the probability of achievingφ should
be maximized, unless an explicit optimization metric is specified for the planning problem. For planning
problems instantiated from a domain declaring the:rewards requirement, the default plan objective is to
maximize the expected reward. A goal statement in the specification of a reward oriented planning problem
identifies a set of absorbing states. In addition to transition rewards specified in action effects, it is possible
to associate a one-time reward for entering a goal state. This is done using the(:goal-reward f )
construct, wheref is a numeric expression.

In general, a statement(:metric maximize f ) in a problem definition means that the expected
value off should be maximized. PPDDL definesgoal-achieved as a special optimization metric, which
can be used to explicitly specify that the plan objective is to maximize (or minimize) the probability of goal
achievement. The value of thegoal-achieved fluent is1 if a goal state has been visited during execution,
and remains0 so long as a goal state has not been visited. The expected value ofgoal-achieved is
therefore equal to the probability of goal achievement.

(define (domain coffee-delivery)
(:requirements :negative-preconditions

:disjunctive-preconditions
:conditional-effects :mdp)

(:predicates (in-office) (raining) (has-umbrella) (is-wet)
(has-coffee) (user-has-coffee))

(:action buy-coffee
:effect (and (when (not (in-office))

(probabilistic 0.8 (has-coffee)))
(when (user-has-coffee)

(increase (reward) 0.8))
(when (not (is-wet))

(increase (reward) 0.2))))
. . . )

Figure 4: Part of PPDDL encoding of “Coffee Delivery” domain.

4 Formal Semantics

We present a formal semantics for PPDDL planning problems in terms of a mapping to a probabilistic
transition system with rewards. A planning problem defines a set of state variablesV , possibly containing
both Boolean and numeric state variables. An assignment of values to state variables defines a state, and
the state spaceS of the planning problem is the set of states representing all possible assignments of values
to variables. In addition toV , a planning problem defines an initial-state distributionp0 : S → [0, 1] with∑

s∈S p0(s) = 1 (i.e. p0 is a probability distribution over states), a formulaφG overV characterizing a set
of goal statesG = {s | s |= φG}, a one-time rewardrG associated with entering a goal state, and a set of
actionsA instantiated from PPDDL action schemata. For goal-directed planning problems, without explicit
rewards, we userG = 1.

4.1 Probability and Reward Structure

An actiona ∈ A consists of a preconditionφa and an effectea. Action a is applicable in a states if and
only if s |= ¬φG ∧φa. It is an error to applya to a state such thats 6|= ¬φG ∧φa. Goal states are absorbing,
so no action may be applied to a state satisfyingφG. The requirement thatφa must hold in order fora to be
applicable is consistent with the semantics of PDDL2.1 (Fox and Long 2003) and permits the modeling of
forced chains of actions. Effects are recursively defined as follows (cf. Rintanen 2003):

1. > is the null-effect, represented in PPDDL by(and) .

2. b and¬b are effects ifb ∈ V is a Boolean state variable.

3. x ← f is an effect ifx ∈ V is a numeric state variable andf is a real-valued function on numeric
state variables.

4. r ↑ f is an effect iff is a real-valued function on numeric state variables.

5. e1 ∧ · · · ∧ en is an effect ife1, . . . , en are effects.

6. c � e is an effect ifc is a formula overV ande is an effect.

7. p1e1| . . . |pnen is an effect ife1, . . . , en are effects,pi ≥ 0 for all i ∈ {1, . . . , n}, and
∑n

i=1 pi = 1.

Items 2 through 4 are referred to assimple effects. The effectb sets the Boolean state variableb to true in the
next state, while¬b setsb to false in the next state. Forx ← f , the value off in the current state becomes
the value of the numeric state variablex in the next state. Effects of the formr ↑ f are used to associate
rewards with transitions as described below.

An actiona = 〈φa, ea〉 defines a transition probability matrixPa and a state reward vectorRa, with
Pa(i, j) being the probability of transitioning to statej when applyinga in statei, andRa(i) being the
expectedreward for executing actiona in statei. We can computePa andRa by first translatingea into an
effect of the formp1e1| . . . |pnen, where eachei is a deterministic effect. The purpose of this translation is
to bring nondeterminism to the top so that we can computePa as

∑n
i=1 piTi, whereTi is a0-1 transition

matrix for ei. Rintanen (2003) calls this form Unary Nondeterminism Normal Form (1ND) and shows that
any effecte can be translated into this form by using the following equivalences:

e ≡1e
e ∧ (p1e1| . . . |pken) ≡p1(e ∧ e1)| . . . |pn(e ∧ en)
c � (p1e1| . . . |pnen) ≡p1(c � e1)| . . . |pn(c � en)

p1(p′1e
′
1| . . . |p′ke′k)|p2e2| . . . |pnen ≡(p1p

′
1)e

′
1| . . . |(p1p

′
k)e

′
k|p2e2| . . . |pnen

The translation into 1ND can result in an exponential increase in the size of the effect formula, although this
representational explosion can be avoided by splitting a single action into multiple actions that are forced to
be executed in sequence. The number of actions that would have to be added is at most linear in the length
of the effect formula.

We further rewrite the effect of an action by translating eachei into an effect of the form(ci1 � ei1) ∧
· · ·∧(cini �eini), where eacheij is a conjunction of simple effects and the conditions are mutually exclusive
and exhaustive (i.e.cij ∧ cik ≡ ⊥ for all j 6= k and

∨ni
j=1 cij ≡ >). The following equivalences allow us to

perform the desired translation:

e ≡>� e

c � e ≡(c � e) ∧ (¬c �>)
c � (c′ � e) ≡(c ∧ c′) � e

(c1 � e1) ∧ (c2 � e2) ≡((c1 ∧ c2) � (e1 ∧ e2)) ∧ ((c1 ∧ ¬c2) � e1)
∧ ((¬c1 ∧ c2) � e2) ∧ ((¬c1 ∧ ¬c2) �>)

This rewrite can also result in an exponential increase in the size of the effect formula, but it will enable us
to succinctly define the transition matrixTi for each deterministic effectei.

An effect of the formc � e, wheree is a conjunction of simple effects, defines a set of state transitions.
We assume thate is consistent, which intuitively means that the constituent parts ofe can be evaluated in
arbitrary order without resulting in a different successor state. Actions with inconsistent effects are not valid
PPDDL actions, and care should be taken when designing a PPDDL domain to ensure that no instantiations
of action schemata can have inconsistent effects. A conjunction of simple effects is inconsistent if it contains
bothb and¬b, or multiplenon-commutativeupdates of a single numeric state variable. Two effectsx ← f
andx← f ′ are commutative iff(s[x = f ′(s)]) = f ′(s[x = f(s)]), wheref(s) is the value off evaluated

in states ands[x = y] denotes a state with all state variables having the same value as in states, except for
x, which has valuey. That is, numeric effects are commutative if they are insensitive to ordering.

Under these assumptions, we define a functionτ that returns a successor state given a state and an effect
formula. The definition ofτ takes two states: the current state and an intermediate successor state. State
changes are accumulated in the second parameter, while the current state is kept unmodified in the first
parameter so that real-valued functions can be evaluated in this state. The definition ofτ is as follows:

τ(s, s′,>) .= s′

τ(s, s′, b) .= s′[b = >]
τ(s, s′,¬b) .= s′[b = ⊥]

τ(s, s′, x← f) .= s′[x = f(s)]
τ(s, s′, r ↑ f) .= s′

τ(s, s′, e1 ∧ e2)
.= τ(s, τ(s, s′, e1), e2)

We can useτ to describe the set of state transitions defined by the effectc � e:

T (c � e) = {〈s, s′〉|s |= c ands′ = τ(s, s, e)}.

Given this definition ofT (c � e), we can compute a transition matrixTij for eachcij � eij . The element
at row s and columns′ of Tij is 1 if 〈s, s′〉 ∈ T (cij � eij), and0 otherwise. Since we have ensured that
the conditionscij are mutually exclusive, we getPa =

∑n
i=1 piTi as the transition probability matrix for

actiona with effect p1e1| . . . |pnen, whereTi =
∑ni

j=1 Tij . To capture the notion of failed preconditions,
we introduce an error states⊥. We set the entry at columns⊥ to 1 for each rows such thats 6|= φa and the
remaining entries of these rows are set to zero. For rowss such thats |= φa, the entry at columns⊥ is set to
zero. Finally, we need to make all states that satisfy the goal conditionφG of the problem absorbing. This
is accomplished by modifyingPa: for eachs such thats |= φG, we set the entry at rows and columns to 1
and the remaining entries on the same row to0.

The reward associated with a conjunction of simple effects can be defined as follows:

r(s,>) .= 0
r(s, b) .= 0

r(s,¬b) .= 0
r(s, x← f) .= 0
r(s, r ↑ f) .= f(s)

r(s, e1 ∧ e2)
.= r(s, e1) + r(s, e2)

The effectcij � eij associates rewardr(s, eij) with each transition〈s, s′〉 ∈ T (cij � eij), and an additional
rewardrG if s′ is a goal state (i.e.s′ |= φG), or no reward ifs is a goal state. We define a transition reward
matrixRij for cij � eij . The element at rows and columns′ of Rij is r(s, eij) if 〈s, s′〉 ∈ T (cij � eij) and
s 6|= φG ands′ 6|= φG, r(s, eij) + rG if 〈s, s′〉 ∈ T (cij � eij) ands 6|= φG ands′ |= φG, and0 otherwise.
We sum over allcij � eij to get a transition reward matrix forei: Ri =

∑ni
j=1 Rij . To obtain the expected

transition reward for an action with effectp1e1| . . . |pnen, we compute the matrixR =
∑n

i=1 piRi. The

elements of the state reward vectorRa can then be computed as follows:Ra(i) =
∑|S|

j=1 R(i, j).

Consider the “Bomb and Toilet” example in Figure 3. This planning problem has four state variables,
and thus a state space of size 16.2 Let b1 be the state variablebomb-in-packagepackage1, b2 the state vari-
ablebomb-in-packagepackage2, b3 the state variabletoilet-clogged , andb4 the state variablebomb-defused .
The actiondunk-packagepackage1 has precondition> (i.e. is applicable in all states) and effect(b1 � b4) ∧
(0.05b3|0.95>). We transform this effect, using the equivalences given above, to the effect0.05((b1 � (b3∧
b4)) ∧ (¬b1 � b3))|0.95((b1 � b4) ∧ (¬b1 �>)). We use the following encoding of states:

b1 b2 b3 b4 state
⊥ ⊥ ⊥ ⊥ 1
⊥ ⊥ ⊥ > 2
⊥ ⊥ > ⊥ 3
...

...
...

...
...

> > > > 16

Given this encoding, we get the following transition probability matrix fordunk-packagepackage1:

Pa =




19
20 0 1

20 0 0 0 0 0 0 0 0 0 0 0 0 0
0 1 0 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 1 0 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 1 0 0 0 0 0 0 0 0 0 0 0 0
0 0 0 0 19

20 0 1
20 0 0 0 0 0 0 0 0 0

0 0 0 0 0 1 0 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 1 0 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 1 0 0 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 19

20 0 1
20 0 0 0 0

0 0 0 0 0 0 0 0 0 1 0 0 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 1 0 0 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 19

20 0 1
20

0 0 0 0 0 0 0 0 0 0 0 0 0 1 0 0
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1
0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 1




Note that all states satisfying the goal condition¬b3 ∧ b4 for the given planning problem have been made
absorbing. The goal states are 2, 6, 10, and 14. We also get the following state reward vector:

Ra = (0, 0, 0, 0, 0, 0, 0, 0,
19
20

, 0, 0, 0,
19
20

, 0, 0, 0)

4.2 Optimality Criteria

We have shown how to construct an MDP from the PPDDL encoding of a planning problem. The plan
objective is to maximizes the expected reward for the MDP. This can be interpreted as expecteddiscounted
reward or expectedtotal reward depending on the situation. For process oriented planning problems (e.g. the
“Coffee Delivery” problem), the former is typically what we want, while the latter often is the interpretation

2Not all 16 states are reachable for this problem. For example, the bomb is in exactly one of the two packages, so
bomb-in-packagepackage1 ≡ ¬bomb-in-packagepackage2 for all states, meaning there are at most 8 reachable states.

chosen for goal oriented problems (e.g. the “Bomb and Toilet” problem). PPDDL can be used to encode
both kind of planning problems, but does not include any facility for enforcing a specific interpretation.

For the probabilistic track of the 4th International Planning Competition, we used expected total reward
as the optimality criterion. This requires some care in the design of planning problems in order to ensure
that the expected total reward is bounded for the optimal policy. The following restrictions were made for
problems used at the planning competition:

1. Each problem had a goal statement, identifying a set of absorbing goal states.

2. A positive reward was associated with transitioning into a goal state.

3. A negative reward (cost) was associated with each action.

4. A “done” action was available in all states, which could be used to end further accumulation of reward.

These conditions ensure that an MDP model of a planning problem is apositive bounded modelPuterman
(1994, pp. 284). The only positive reward is for transitioning into a goal state. Since goal states are absorbing
(i.e. they have no outgoing transitions), the maximum value for any state is bounded by the goal reward.
Furthermore, the “done” action ensures that there is an action available in each state which guarantees a
non-negative future reward.

5 Conversion to Dynamic Bayesian Network

In this section, we focus on PPDDL problems without numeric state variables as such problems are guar-
anteed to have a finite state space. For a problem withn Boolean state variables, the transition probability
matrix for each action may contain up to22n entries. We showed how to compute these entries in the
previous section, but the computation relied on transformations of action effects that could result in an ex-
ponential increase in the representation size compared to the original PPDDL encoding. We will now show
how to compute afactoredrepresentation of the transition probability matrix for a PPDDL action without
more than a polynomial increase in size. The transition probability matrix will be represented using ady-
namic Bayesian network(DBN; Dean and Kanazawa 1989), whose structure can be exploited by algorithms
for decision theoretic planning (see, e.g., work by Boutilier et al. 1995; Hoey et al. 1999; Boutilier et al.
1999; Guestrin et al. 2003).

A Bayesian network is a directed graph. Each node of the graph represents a state variable, and a
directed edge from one node to another represents a causal dependence. With each node is associated a
conditional probability table (CPT). The CPT for state variableX ’s node represents the probability distribu-
tion over possible values forX conditioned on the values of state variables whose nodes are parents ofX ’s
node. A Bayesian network is a factored representation of the joint probability distribution over the variables
represented in the network.

A DBN is a Bayesian network with a specific structure aimed at capturing temporal dependence. For
each state variableX, we create a duplicate state variableX ′, with X representing the situation at the present
time andX ′ representing the situation one time step into the future. A directed edge from a present-time state
variableX to a future-time state variableY ′ encodes a temporal dependence. There are no edges between
two present-time state variables, or from a future-time to a present-time state variable (the present does not
depend on the future). We can, however, have an edge between two future-time state variables. Such edges,
calledsynchronicedges, are used to represent correlated effects. A DBN is a factored representation of the

joint probability distribution over present-time and future-time state variables, which is also the transition
probability matrix for a discrete-time Markov process.

We now show how to generate a DBN representing the transition probability matrix for a PPDDL action.
To avoid representational blowup, we introduce a multi-valued auxiliary variable for each probabilistic effect
of an action effect. These auxiliary variables are used to indicate which of the possible outcomes of a
probabilistic effect that occurs, and this allows us to correlate all the effects of a specific outcome. The
auxiliary variable associated with a probabilistic effect withn outcomes can take onn different values. A
PPDDL effecte of size |e| can consist of at mostO(|e|) distinct probabilistic effects. Hence, the number
of auxiliary variables required to encode the transition probability matrix for an action with effecte will be
at mostO(|e|). Only future-time versions of the auxiliary variables are necessary. For a PPDDL problem
with m Boolean state variables, we need on the order of2m+maxa∈A |ea| nodes in the DBNs representing
transition probability matrices for actions.

We provide a compositional approach for generating a DBN that represents the transition probability
matrix for a PPDDL action with preconditionφa and effectea. We assume that the effect is consistent, i.e.
thatb and¬b do not occur in the same outcome with overlapping conditions. The DBN for an empty effect
> simply consists of2m nodes, with each present-time nodeX connected to its future-time counterpartX ′.
The CPT forX ′ has the non-zero entriesPr[X ′ = > | X = >] = 1 andPr[X ′ = ⊥ | X = ⊥] = 1. The
same holds for a reward effectr ↑ k, which does not change the value of state variables.

Next, consider the simple effectsb and¬b. LetXb be the state variable associated with the PPDDL atom
b. For these effects, we eliminate the edge fromXb to X ′

b. The CPT forX ′
b has the entryPr[X ′

b = >] = 1
for effectb andPr[X ′

b = ⊥] = 1 for effect¬b.
For conditional effects,c � e, we take the DBN fore and add edges between the present-time state

variables mentioned in the formulac and the future-time state variables in the DBN fore. Entries in the
CPT for a state variableX ′ that correspond to settings of the present-time state variables that satisfyc
remain unchanged. The other entries are set to1 if X is true and0 otherwise (the value ofX does not
change if the effect condition is not satisfied).

The DBN for an effect conjunctione1∧· · ·∧en is constructed from the DBNs for then effect conjuncts.
The value forPr[X ′ = > | X] in the DBN for the conjunction is set to the maximum ofPr[X ′ = > | X]
over the DBNs for the conjuncts.

Finally, to construct a DBN for a probabilistic effectp1e1| . . . |pnen, we introduce an auxiliary variable
Y ′ that is used to indicate which one of then outcomes occurred. The node forY ′ does not have any parents,
and the entries of the CPT arePr[Y ′ = i] = pi. Given a DBN forei, we add a synchronic edge fromY ′

to all state variablesX. The value ofPr[X ′ = > | X, Y ′ = j] is set toPr[X ′ = > | X] if j = i and0
otherwise. We do this for alln outcomes, which results inn DBNs. These DBNs are combined in the same
way as for conjunctive effects, and the result is the DBN for the probabilistic effect.

The process of constructing a DBN from a PPDDL encoding of an action is illustrated in Figure 5 for
the “Bomb and Toilet” example. Note that the CPTs for state variables only contain0 and1 entries. The
probabilities of different outcomes are encoded in the CPTs for auxiliary variables.

The effects in the “Bomb and Toilet” example are fairly simple. Figure 6(a) shows the PPDDL encoding
for an action in the “Coffee Delivery” domain. The effect of this action has both nested probabilistic effects
and correlated effects (the first outcome of the first probabilistic effect is an example of the latter). The struc-
ture of the DBN for a this more complex example is shown in Figure 6(b). There are three auxiliary variables
because the action effect contains three probabilistic effects. The node labeledHC ′ (the future-time version
of the state variablehas-coffee) has five parents, including all three auxiliary variables. Consequently, the
CPT for this node will have25 = 32 rows.

BIP2 BIP′2

BIP1 BIP′1

BD BD′

TC TC′

(a)BD

BIP2 BIP′2

BIP1 BIP′1

BD BD′

TC TC′

(b) TC

BIP2 BIP′2

BIP1 BIP′1

BD BD′

TC TC′

(c) BIP1 � BD

BIP2 BIP′2

BIP1 BIP′1

BD BD′

TC TC′

Aux′

(d) 0.05TC |0.95>

BIP2 BIP′2

BIP1 BIP′1

BD BD′

TC TC′

Aux′

(e)(BIP1�BD)∧(0.05TC |0.95>)

BIP ′
1

BIP1 > ⊥
> 1 0
⊥ 0 1

BIP ′
2

BIP2 > ⊥
> 1 0
⊥ 0 1

BD ′

BIP1 BD > ⊥
> > 1 0
> ⊥ 1 0
⊥ > 1 0
⊥ ⊥ 0 1

TC ′

TC Aux ′ > ⊥
> 1 1 0
> 2 1 0
⊥ 1 1 0
⊥ 2 0 1

Aux ′

1 2

0.05 0.95

Figure 5: Construction of a DBN for thedunk-packagepackage1 action of the bomb and toiled example. The final
DBN is shown in (e) and its CPTs are listed at the bottom of the figure.

(:action deliver-coffee
:effect (and (when (and (in-office) (has-coffee))

(probabilistic 0.8 (and (user-has-coffee)
(not (has-coffee))
(increase (reward) 0.8))

0.2 (and (probabilistic 0.5 (not (has-coffee)))
(when (user-has-coffee)

(increase (reward) 0.8)))))
(when (and (not (in-office)) (has-coffee))

(and (probabilistic 0.8 (not (has-coffee)))
(when (user-has-coffee)

(increase (reward) 0.8))))
(when (and (not (has-coffee)) (user-has-coffee))

(increase (reward) 0.8))
(when (not (is-wet))

(increase (reward) 0.2))))

(a)

R R′

HU HU ′

IW IW′

UHC UHC′ Aux′1

HC HC′ Aux′2

IO IO′ Aux′3

(b)

Figure 6: PPDDL encoding (a) and DBN (b) for the “deliver-coffee” action of the “Coffee Delivery” domain.

A BNF Grammar for PPDDL1.0

We provide the full syntax for PPDDL1.0 using an extended BNF notation with the following conventions:

• Each rule is of the form〈non-terminal〉 ::= expansion.

• Alternative expansions are separated by a vertical bar (“|”).
• A syntactic element surrounded by square brackets (“[“ and “]”) is optional.

• Expansions and optional syntactic elements with a superscripted requirements flag are only available
if the requirements flag is specified for the domain or problem currently being defined. For example,
[〈types-def〉]:typing in the syntax for domain definitions means that〈types-def〉 may only occur in
domain definitions that include the:typing flag in the requirements declaration.

• An asterisk (“*”) following a syntactic elementx means zero or more occurrences ofx; a plus (“+”)
following x means at least one occurrence ofx.

• Parameterized non-terminals, for example〈typed list(x)〉, represent separate rules for each instantia-
tion of the parameter.

• Terminals are written usingtypewriter font.

• The syntax is Lisp-like. In particular this means that case is not significant (e.g.?x and?X are equiv-
alent), parenthesis are an essential part of the syntax and have no semantic meaning in the extended
BNF notation, and any number of whitespace characters (space, newline, tab, etc.) may occur between
tokens.

A.1 Domains

The syntax for domain definitions is the same as for PDDL2.1, except that durative actions are not allowed.
Declarations of constants, predicates, and functions are allowed in any order with respect to one another, but
they must all come after any type declarations and precede any action declarations.

〈domain〉 ::= ( define ( domain 〈name〉 )
[〈require-def〉]
[〈types-def〉]:typing

[〈constants-def〉]
[〈predicates-def〉]
[〈functions-def〉]:fluents

〈structure-def〉* )
〈require-def〉 ::= ( :requirements 〈require-key〉* )
〈require-key〉 ::= See Section A.4
〈types-def〉 ::= ( :types 〈typed list(name)〉 )
〈constants-def〉 ::= ( :constants 〈typed list(name)〉 )
〈predicates-def〉 ::= ( :predicates 〈atomic formula skeleton〉* )
〈atomic formula skeleton〉 ::= ( 〈predicate〉 〈typed list(variable)〉 )
〈predicate〉 ::= 〈name〉
〈functions-def〉 ::= ( :functions 〈function typed list(function skeleton)〉 )

〈function skeleton〉 ::= ( 〈function symbol〉 〈typed list(variable)〉 )
〈function symbol〉 ::= 〈name〉
〈structure-def〉 ::= 〈action-def〉
〈action-def〉 ::= See Section A.2
〈typed list(x)〉 ::= 〈x〉* |:typing 〈x〉+ - 〈type〉 〈typed list(x)〉
〈type〉 ::= ( either 〈primitive type〉+ ) | 〈primitive type〉
〈primitive type〉 ::= 〈name〉
〈function typed list(x)〉 ::= 〈x〉* | :typing 〈x〉+ - 〈function type〉 〈function typed list(x)〉
〈function type〉 ::= number

A 〈name〉 is a string of characters starting with an alphabetic character followed by a possibly empty se-
quence of alphanumeric characters, hyphens (“- ”), and underscore characters (“”). A 〈variable〉 is a〈name〉
immediately preceded by a question mark (“?”). For example,in-office andball 2 are names, and
?gripper is a variable.

A.2 Actions

Action definitions and goal descriptions have the same syntax as in PDDL2.1.

〈action-def〉 ::= ( :action 〈action symbol〉
[:parameters ( 〈typed list(variable)〉 ) ]
〈action-def body〉 )

〈action symbol〉 ::= 〈name〉
〈action-def body〉 ::= [:precondition 〈GD〉]

[:effect 〈effect〉]
〈GD〉 ::= 〈atomic formula(term)〉 | ( and 〈GD〉* )

|:equality ( = 〈term〉 〈term〉 )
|:equality ( not ( = 〈term〉 〈term〉 ) )
|:negative-preconditions ( not 〈atomic formula(term)〉 )
|:disjunctive-preconditions ( not 〈GD〉 )
|:disjunctive-preconditions ( or 〈GD〉* )
|:disjunctive-preconditions ( imply 〈GD〉 〈GD〉 )
|:existential-preconditions ( exists ( 〈typed list(variable)〉 ) 〈GD〉 )
|:universal-preconditions ( forall ( 〈typed list(variable)〉 ) 〈GD〉 )
|:fluents 〈f-comp〉

〈atomic formula(x)〉 ::= ( 〈predicate〉 〈x〉* ) | 〈predicate〉
〈term〉 ::= 〈name〉 | 〈variable〉
〈f-comp〉 ::= ( 〈binary-comp〉 〈f-exp〉 〈f-exp〉 )
〈binary-comp〉 ::= < | <= | = | >= | >
〈f-exp〉 ::= 〈number〉 | 〈f-head(term)〉

| ( 〈binary-op〉 〈f-exp〉 〈f-exp〉 ) | ( - 〈f-exp〉 )
〈f-head(x)〉 ::= ( 〈function symbol〉 〈x〉* ) | 〈function symbol〉
〈binary-op〉 ::= + | - | * | /

A 〈number〉 is a sequence of numeric characters, possibly with a single decimal point (“. ”) at any position
in the sequence. Negative numbers are written as(- 〈number〉) .

The syntax for effects has been extended to allow for probabilistic effects, which can be arbitrarily
interleaved with conditional effects and universal quantification.

〈effect〉 ::= 〈p-effect〉 | ( and 〈effect〉* )
|:conditional-effects ( forall ( 〈typed list(variable)〉 ) 〈effect〉 )
|:conditional-effects ( when 〈GD〉 〈effect〉 )
|:probabilistic-effects ( probabilistic 〈prob-effect〉+ )

〈p-effect〉 ::= 〈atomic formula (term)〉 | ( not 〈atomic formula(term)〉 )
|:fluents ( 〈assign-op〉 〈f-head(term)〉 〈f-exp〉 )
|:rewards ( 〈additive-op〉 〈reward fluent〉 〈f-exp〉 )

〈prob-effect〉 ::= 〈probability〉 〈effect〉
〈assign-op〉 ::= assign | scale-up | scale-down | 〈additive-op〉
〈additive-op〉 ::= increase | decrease
〈reward fluent〉 ::= ( reward ) | reward

A 〈probability〉 is a〈number〉 with a value in the interval[0, 1].

A.3 Problems

The syntax for problem definitions has been extended to allow for the specification of a probability distribu-
tion over initial states, and also to permit the association of a one-time reward with entering a goal state. It
is otherwise identical to the syntax for PDDL2.1 problem definitions.

〈problem〉 ::= ( define ( problem 〈name〉 )
( :domain 〈name〉 )
[〈require-def〉]
[〈objects-def〉]
[〈init〉]
〈goal〉 )

〈objects-def〉 ::= ( :objects 〈typed list(name)〉 )
〈init〉 ::= ( :init 〈init-el〉* )
〈init-el〉 ::= 〈p-init-el〉

|:probabilistic-effects ( probabilistic 〈prob-init-el〉* )
〈p-init-el〉 ::= 〈atomic formula(name)〉 |:fluents ( = 〈f-head(name)〉 〈number〉 )
〈prob-init-el〉 ::= 〈probability〉 〈a-init-el〉
〈a-init-el〉 ::= 〈p-init-el〉 | ( and 〈p-init-el〉* )
〈goal〉 ::= 〈goal-spec〉 [〈metric-spec〉] | 〈metric-spec〉
〈goal-spec〉 ::= ( :goal 〈GD〉 ) [( :goal-reward 〈ground-f-exp〉 ) ]:rewards

〈metric-spec〉 ::= ( :metric 〈optimization〉 〈ground-f-exp〉 )
〈optimization〉 ::= minimize | maximize
〈ground-f-exp〉 ::= 〈number〉 | 〈f-head(name)〉

| ( 〈binary-op〉 〈ground-f-exp〉 〈ground-f-exp〉 )
| ( - 〈ground-f-exp〉 )
| ( total-time ) | total-time
| ( goal-achieved ) | goal-achieved
|:rewards 〈reward fluent〉

A.4 Requirements

Below is a table of all requirements in PPDDL1.0. Some requirements imply others; some are abbrevia-
tions for common sets of requirements. If a domain stipulates no requirements, it is assumed to declare a
requirement for:strips .

Requirement Description
:strips Basic STRIPS-style adds and deletes
:typing Allow type names in declarations of variables
:equality Support= as built-in predicate
:negative-preconditions Allow negated atoms in goal descriptions
:disjunctive-preconditions Allow disjunctive goal descriptions
:existential-preconditions Allow exists in goal descriptions
:universal-preconditions Allow forall in goal descriptions
:quantified-preconditions = :existential-preconditions

+ :universal-preconditions
:conditional-effects Allow when andforall in action effects
:probabilistic-effects Allow probabilistic in action effects
:rewards Allow reward fluent in action effects and

optimization metric
:fluents Allow numeric state variables
:adl = :strips + :typing + :equality

+ :negative-preconditions
+ :disjunctive-preconditions
+ :quantified-preconditions
+ :conditional-effects

:mdp = :probabilistic-effects + :rewards

Åström, K. J. 1965. Optimal control of Markov processes with incomplete state information.Journal of
Mathematical Analysis and Applications10, no. 1: 174–205.

Boutilier, Craig, Thomas Dean, and Steve Hanks. 1999. Decision-theoretic planning: Structural assump-
tions and computational leverage.Journal of Artificial Intelligence Research11: 1–94.

Boutilier, Craig, Richard Dearden, and Mois´es Goldszmidt. 1995. Exploiting structure in policy construc-
tion. In Proceedings of the Fourteenth International Joint Conference on Artificial Intelligence, edited
by Chris S. Mellish, 1104–1111, Montreal, Canada. Morgan Kaufmann Publishers.

Dean, Thomas and Keiji Kanazawa. 1989. A model for reasoning about persistence and causation.Com-
putational Intelligence5, no. 3: 142–150.

Dearden, Richard and Craig Boutilier. 1997. Abstraction and approximate decision-theoretic planning.
Artificial Intelligence89, no. 1–2: 219–283.

Draper, Denise, Steve Hanks, and Daniel S. Weld. 1994. Probabilistic planning with information gath-
ering and contingent execution. InProceedings of the Second International Conference on Artificial
Intelligence Planning Systems, edited by Kristian Hammond, 31–36, Chicago, IL. AAAI Press.

Fox, Maria and Derek Long. 2003. PDDL2.1: An extension to PDDL for expressing temporal planning
domains.Journal of Artificial Intelligence Research20: 61–124.

Ghallab, Malik, Adele E. Howe, Craig A. Knoblock, Drew McDermott, Ashwin Ram, Manuela M. Veloso,
Daniel S. Weld, and David Wilkins. 1998. PDDL—the planning domain definition language. Tech-
nical Report CVC TR-98-003/DCS TR-1165, Yale Center for Computational Vision and Control, New
Haven, CT.

Guestrin, Carlos, Daphne Koller, Ronald Parr, and Shobha Venkataraman. 2003. Efficient solution algo-
rithms for factored MDPs.Journal of Artificial Intelligence Research19: 399–468.

Hoey, Jesse, Robert St-Aubin, Alan Hu, and Craig Boutilier. 1999. SPUDD: Stochastic planning using
decision diagrams. InProceedings of the Fifteenth Conference on Uncertainty in Artificial Intelligence,
edited by Kathryn B. Laskey and Henri Prade, 279–288, Stockholm, Sweden. Morgan Kaufmann Pub-
lishers.

Howard, Ronald A. 1960.Dynamic Programming and Markov Processes. New York, NY: John Wiley &
Sons.

. 1971. Dynamic Probabilistic Systems, vol. II: Semi-Markov and Decision Processes. New York,
NY: John Wiley & Sons.

Kushmerick, Nicholas, Steve Hanks, and Daniel S. Weld. 1995. An algorithm for probabilistic planning.
Artificial Intelligence76, no. 1–2: 239–286.

Littman, Michael L. 1997. Probabilistic propositional planning: Representations and complexity. In
Proceedings of the Fourteenth National Conference on Artificial Intelligence, 748–754, Providence,
RI. AAAI Press.

Mausam and Daniel S. Weld. 2004. Solving concurrent Markov decision processes. InProceedings of the
Nineteenth National Conference on Artificial Intelligence, 716–722, San Jose, CA. AAAI Press.

McDermott, Drew. 2000. The 1998 AI planning systems competition.AI Magazine21, no. 2: 35–55.

20



Puterman, Martin L. 1994.Markov Decision Processes: Discrete Stochastic Dynamic Programming. New
York, NY: John Wiley & Sons.

Rintanen, Jussi. 2003. Expressive equivalence of formalism for planning with sensing. InProceedings
of the Thirteenth International Conference on Automated Planning and Scheduling, edited by Enrico
Giunchiglia, Nicola Muscettola, and Dana S. Nau, 185–194, Trento, Italy. AAAI Press.

